In[291]:=

Mathematical Economics 2020 – 2021

Optimal control in continuous time

Paulo Brito 18 / 12 / 2020

(*

Problem 2.2.1

Consumption-investment problem

Finite horizon, initial state given A(0) = A0, terminal state given A(T) = A0

Observation: the current-value Hamiltonian is used in the solution

*)

In[294]:=

```
(* definition of the functions *)
ClearAll[Cs, A, As, TT, Q, r, ρ, A0]
f[Cs_] := Log[Cs]
g[A_, Cs_] := r A - Cs
h[Cs_, A_, Q_] := f[Cs] + Q g[A, Cs]
```

In[298]:=

```
(* optimality condition *)
Solve[D[h[Cs, A, Q], Cs] == 0, Cs]
```

Out[298]= $\left\{\left\{Cs \to \frac{1}{Q}\right\}\right\}$

In[299]:=

```
(* solution of the generalized MHDS *)
mhds221 =
 DSolve[{D[Q[t], t] == (ρ - r) Q[t], D[A[t], t] == g[A[t], 1 / Q[t]]}, {A[t], Q[t]}, t]
Qs[t_] := Evaluate[Q[t] /. mhds221]
As[t_] := Evaluate[A[t] /. mhds221]
```

Out[299]= $\left\{\left\{Q[t] \to e^{-r t + t \rho}\, c_1,\ A[t] \to \frac{e^{r t - t \rho}}{\rho\, c_1} + e^{r t}\, c_2\right\}\right\}$

In[302]:=
```
(* Obtaining the particular solution, that is, the solution to the problem *)
Ps221 := Assuming[TT > 0, Simplify[Solve[{As[0] == A0, As[TT] == A0}, {C[1], C[2]}]]]
Csol[t_, TT_, A0_, ρ_, r_] = 1 / Evaluate[Qs[t] /. Ps221]
Asol[t_, TT_, A0_, ρ_, r_] = Simplify[Evaluate[As[t] /. Ps221]]
```

Out[303]= $\left\{\left\{\frac{A0\, e^{r\,t-TT\,(r-\rho)-t\,\rho}\left(-1+e^{r\,TT}\right)\rho}{-1+e^{TT\,\rho}}\right\}\right\}$

Out[304]= $\left\{\left\{-\frac{A0\, e^{(t-TT)\,(r-\rho)}\left(1-e^{r\,TT}-e^{t\,\rho}+e^{r\,TT+t\,\rho-TT\,\rho}\right)}{-1+e^{TT\,\rho}}\right\}\right\}$

In[305]:=
```
(* Optimal trajectories for asset holdings and for consumption:
    for A0= 1, T = 10, ρ=0.02 and r=0.025  *)
Plot[{Asol[t, 10, 1, 0.02, 0.025]}, {t, 0, 10}, AxesLabel → {"t", "A(t)"}]
Plot[{Csol[t, 10, 1, 0.02, 0.025]}, {t, 0, 10}, AxesLabel → {"t", "C(t)"}]
```

Out[305]=


Out[306]=


In[307]:=

(*

Problem 2.2.2

Consumption–investment problem

Finite horizon,initial state given A(0)=A0,terminal state constrained A(T)≥0

Observation:the current–value Hamiltonian is used in the solution

*)

In[308]:=
```
(* definition of the functions *)
ClearAll[Cs, A, As, TT, Q, r, ρ, A0]
f[Cs_] := Log[Cs]
g[A_, Cs_] := r A - Cs
h[Cs_, A_, Q_] := f[Cs] + Q g[A, Cs]
```

In[312]:=
```
(* optimality condition *)
Solve[D[h[Cs, A, Q], Cs] == 0, Cs]
```

Out[312]= $\left\{\left\{Cs \to \frac{1}{Q}\right\}\right\}$

In[313]:=
```
(* mhds and solution of the generalized MHDS *)
mhds222 =
 DSolve[{D[Q[t], t] == (ρ - r) Q[t], D[A[t], t] == g[A[t], 1 / Q[t]]}, {A[t], Q[t]}, t]
Qs[t_] := Evaluate[Q[t] /. mhds222]
As[t_] := Evaluate[A[t] /. mhds222]
```

Out[313]= $\left\{\left\{Q[t] \to e^{-r\,t+t\,\rho}\,c_1,\ A[t] \to \frac{e^{r\,t-t\,\rho}}{\rho\,c_1} + e^{r\,t}\,c_2\right\}\right\}$

In[316]:=
```
(* Obtaining the particular solution, that is, the solution to the problem
*)
Ps222 := Assuming[TT > 0,
  Simplify[Solve[{As[0] == A0, Exp[-ρ TT] Qs[TT] × As[TT] == 0}, {C[1], C[2]}]]]
Csol[t_, TT_, A0_, ρ_, r_] = 1 / Evaluate[Qs[t] /. Ps222]
Asol[t_, TT_, A0_, ρ_, r_] = Simplify[Evaluate[As[t] /. Ps222]]
```

Out[317]= $\left\{\left\{\frac{A0\,e^{r\,t-t\,\rho}\,\rho}{1-e^{-TT\,\rho}}\right\}\right\}$

Out[318]= $\left\{\left\{\frac{A0\,e^{r\,t}\left(-1+e^{(-t+TT)\,\rho}\right)}{-1+e^{TT\,\rho}}\right\}\right\}$

In[319]:=
```
(* Optimal trajectories for asset holdings and for consumption:
   for A0= 1, T = 20, ρ=0.02 and r=0.025  *)
Plot[{Asol[t, 20, 1, 0.02, 0.025]}, {t, 0, 20}, AxesLabel → {"t", "A(t)"}]
Plot[{Csol[t, 20, 1, 0.02, 0.025]}, {t, 0, 20}, AxesLabel → {"t", "C(t)"}]
```

Out[319]=



Out[320]=



In[321]:= (*

Problem 2.2.9

Firm's investment problem

Infinite horizon, initial state given K(0)=k0, free terminal state

*)

In[322]:=
```
ClearAll[Q, Qs, IV, K, Ks, K0, t, TT, r, p, q, δ]
f[K_, IV_] := p K  - q IV^2
g[K_, IV_] := IV - δ K
H[IV_, K_, Q_] = f[K, IV] + Q g[K, IV]
```

Out[325]= $K\,p - IV^2\,q + Q\,(IV - K\,\delta)$

In[326]:=
```
(* optimal investment *)
oc229 = Solve[D[H[IV, K, Q], IV] == 0, IV]
```

Out[326]= $\left\{\left\{IV \to \frac{Q}{2\,q}\right\}\right\}$

In[327]:=
```
IVf[Q_] = Evaluate[IV /. oc229[[1]]]
```
Out[327]= $\frac{Q}{2\,q}$

In[328]:=
```
HK[Q_] = D[H[IV, K, Q], K]
```
Out[328]= $p - Q\,\delta$

In[329]:=

In[330]:=
```
(* MHDS and generalized solution *)
mhds229 = DSolve[{D[Q[t], t] == r Q[t] - HK[Q[t]],
   D[K[t], t] == g[K[t], IVf[Q[t]]]}, {K[t], Q[t]}, t]
Qs[t_] := Evaluate[Q[t] /. mhds229]
Ks[t_] := Evaluate[K[t] /. mhds229]
```
Out[330]= $$\left\{\left\{K[t] \to \frac{e^{-t\,\delta - t\,(r+\delta)}\left(-1 + e^{t\,\delta + t\,(r+\delta)}\right) p}{2\,q\,(r+\delta)\,(r+2\,\delta)} + \frac{e^{-t\,\delta}\left(-\frac{e^{t\,(-r-\delta)}\,p}{-r-\delta} + \frac{e^{t\,\delta}\,p}{\delta}\right)}{2\,q\,r + 4\,q\,\delta} + e^{-t\,\delta}\,c_1 + \frac{e^{-t\,\delta}\left(-1 + e^{t\,\delta + t\,(r+\delta)}\right) c_2}{2\,q\,(r+2\,\delta)},\ Q[t] \to \frac{p}{r+\delta} + e^{t\,(r+\delta)}\,c_2\right\}\right\}$$

In[333]:=
```
(* transversality condition*)
TC229[t_] = Simplify[Exp[-r t] Qs[t]]
```
Out[333]= $$\left\{e^{-r\,t}\left(\frac{p}{r+\delta} + e^{t\,(r+\delta)}\,c_2\right)\right\}$$

In[334]:=
```
(* Solution to the problem: particular solution *)
PS229 = Simplify[Solve[{Ks[0] == K0, TC229[TT] == 0}, {C[1], C[2]}]]
```
Out[334]= $$\left\{\left\{c_1 \to K0 - \frac{p}{2\,q\,r\,\delta + 2\,q\,\delta^2},\ c_2 \to -\frac{e^{-TT\,(r+\delta)}\,p}{r+\delta}\right\}\right\}$$

In[335]:=
```
Qsol = Evaluate[Qs[t] /.  PS229[[1]]]
Ivsol[t_, TT_, K0_, r_, δ_, p_, q_] =  IVf[Qsol]
Ksol[t_, TT_, K0_, r_, δ_, p_, q_] = Simplify[Evaluate[Ks[t] /. PS229[[1]]]]
```
Out[335]= $$\left\{\frac{p}{r+\delta} - \frac{e^{t\,(r+\delta) - TT\,(r+\delta)}\,p}{r+\delta}\right\}$$

Out[336]= $$\left\{\frac{\frac{p}{r+\delta} - \frac{e^{t\,(r+\delta) - TT\,(r+\delta)}\,p}{r+\delta}}{2\,q}\right\}$$

Out[337]= $$\left\{\frac{1}{2\,q\,\delta\,(r+\delta)\,(r+2\,\delta)}\,e^{-r\,TT - (t+TT)\,\delta}\left(p\,\delta - e^{t\,(r+2\,\delta)}\,p\,\delta + e^{r\,TT + (t+TT)\,\delta}\,p\,(r+2\,\delta) + e^{TT\,(r+\delta)}\,(r+2\,\delta)\,(-p + 2\,K0\,q\,\delta\,(r+\delta))\right)\right\}$$

In[338]:=
```
(* Optimal trajectories for the capital stock and investment:
   for K0= 1, T = 50, δ=0.05, p=1, q=1/2 and r=0.025  *)
Plot[{Ksol[t, 50, 1, 0.025, 0.05, 1, 0.5]}, {t, 0, 50}, AxesLabel → {"t", "K(t)"}]
Plot[{Ivsol[t, 50, 1, 0.025, 0.05, 1, 0.5]}, {t, 0, 50}, AxesLabel → {"t", "I(t)"}]
```

Out[338]=


Out[339]=


In[340]:= (*

Problem 2.2.11

Endogenous growth model

Infinite horizon, initial state given A(0)=A0,

terminal state constrained lim_t → ∞ e^(-rt) K(T)≥0

*)

In[341]:=
```
(* definition of the functions *)
ClearAll[Cs, K, Q, A, Ks, Qs, ρ, θ, K0]
f[Cs_] := (Cs^(1 - θ) - 1) / (1 - θ)
g[K_, Cs_] := A K - Cs
h[Cs_, K_, Q_] := f[Cs] + Q g[K, Cs]
```

In[345]:=
```
Solve[D[h[Cs, K, Q], Cs] == 0, Cs]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[345]= $\{\{Cs \to Q^{-1/\theta}\}\}$

In[346]:=
```
(* MHDS and generalized solution *)
mhds2211 =
 DSolve[{D[Q[t], t] == (ρ - A) Q[t], D[K[t], t] == g[K[t], Q[t]^(-1/θ)]}, {K[t], Q[t]}, t]
Qs[t_] := Evaluate[Q[t] /. mhds2211]
Ks[t_] := Evaluate[K[t] /. mhds2211]
```

Out[346]= $\left\{\left\{Q[t] \to e^{-A\,t+t\,\rho}\,c_1,\ K[t] \to \frac{\theta\,\left(e^{t\,(-A+\rho)}\,c_1\right)^{-1/\theta}}{A\,(-1+\theta)+\rho} + e^{A\,t}\,c_2\right\}\right\}$

In[349]:=
```
(* transversality condition*)
TC2211[t_] = Simplify[Exp[-ρ t] Qs[t]×Ks[t]]
```

Out[349]= $\left\{c_1\left(\frac{e^{-A\,t}\,\theta\,\left(e^{t\,(-A+\rho)}\,c_1\right)^{-1/\theta}}{A\,(-1+\theta)+\rho} + c_2\right)\right\}$

In[350]:=
```
(* Solution to the problem: particular solution *)
PS2211 = Simplify[Assuming[ θ > 0 && ρ > 0 && A > (A - ρ) / θ && K0 > 0,
   Solve[{Ks[0] == K0, Limit[TC2211[t], t → Infinity] == 0}, {C[1], C[2]}]]]
```

Solve: Inconsistent or redundant transcendental equation. After reduction, the bad equation is $-\left(c_1^{\frac{1}{\theta}}\right)^{\theta}$ == 0.

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[350]= $\left\{\left\{c_1 \to \left(\frac{\theta}{K0\,(A\,(-1+\theta)+\rho)}\right)^{\theta},\ c_2 \to 0\right\},\ \left\{c_1 \to 0,\ c_2 \to \frac{A\,K0\,(-1+\theta) - 0^{-1/\theta}\,\theta + K0\,\rho}{A\,(-1+\theta)+\rho}\right\}\right\}$

In[351]:=
```
(* we take the first solution*)
Csol[t_, TT_, K0_, ρ_, A_, θ_] = Evaluate[Qs[t] /. PS2211[[1]]] ^ (-1 / θ)
Ksol[t_, TT_, K0_, ρ_, A_, θ_] = Factor[Evaluate[Ks[t] /. PS2211[[1]]]]
```

Out[351]= $\left\{\left(e^{-A\,t+t\,\rho}\left(\frac{\theta}{K0\,(A\,(-1+\theta)+\rho)}\right)^{\theta}\right)^{-1/\theta}\right\}$

Out[352]= $\left\{\frac{\theta\,\left(e^{t\,(-A+\rho)}\left(\frac{\theta}{K0\,(A\,(-1+\theta)+\rho)}\right)^{\theta}\right)^{-1/\theta}}{-A + A\,\theta + \rho}\right\}$

In[353]:=
```
Ksol[t, 20, 3, 0.02, 0.025, 2]
```

Out[353]= $\left\{\frac{3.}{\sqrt{e^{-0.005\,t}}}\right\}$

In[354]:=
```
(* Optimal trajectories for asset holdings and for consumption:
   for K0= 3, T = 20, ρ=0.02, A=0.007 and θ=2  *)
Plot[{N[Ksol[t, 50, 3, 0.02, 0.07, 2]]}, {t, 0, 50}, AxesLabel → {"t", "K(t)"}]
Plot[{N[Csol[t, 50, 3, 0.02, 0.07, 2]]}, {t, 0, 50}, AxesLabel → {"t", "C(t)"}]
```

Out[354]=



Out[355]=



In[356]:= (*

Problem 2.2.14

*)

In[357]:=
```
ClearAll[q]
H[Cv_, Z_, q_] := Log[(Cv^β) (Z^(1 - β))] + q * (δ (Z - Cv))
qs = Solve[D[H[Cv, Z, q], Cv] == 0, q]
D[H[Cv, Z, q], Z]
```

Out[359]= $\left\{\left\{q \to \frac{\beta}{Cv\ \delta}\right\}\right\}$

Out[360]= $\frac{1-\beta}{Z} + q\ \delta$

In[361]:=
```
sol2 = DSolve[{D[C[t], t] == (δ - ρ) C[t] + ((1 - β) δ / β ) C[t]^2 / Z[t],
   D[Z[t], t] == δ (Z[t] - C[t]), Z[0] == Z0}, {Z[t], C[t]}, t]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[361]=
$$\left\{\left\{C_t \to e^{t(\delta-\beta\rho)}\, Z0\, \beta\rho\, (\delta+\beta\rho\, c_1)^{-\beta}\left(\delta + e^{t\rho}\beta\rho\, c_1\right)^{-1+\beta},\; Z[t] \to e^{t(\delta-\beta\rho)}\, Z0\, (\delta+\beta\rho\, c_1)^{-\beta}\left(\delta+e^{t\rho}\beta\rho\, c_1\right)^{\beta}\right\}\right\}$$

In[362]:=
```
C1x = Solve[Limit[Evaluate[(β / δ) Exp[- ρ t] (Z[t] / C[t]) /. sol2],
   t → Infinity, Assumptions → ρ > 0] == 0, C[1]]
```

Out[362]= $\{\{c_1 \to 0\}\}$

In[363]:=
```
Cs[t_] := Simplify[C[t] /. sol2  /. C[1] → 0]
Zs[t_] := Simplify[Z[t] /. sol2  /. C[1] → 0]
Cs[t]
Zs[t]
```

Out[365]= $\left\{\dfrac{e^{t(\delta-\beta\rho)}\, Z0\, \beta\rho}{\delta}\right\}$

Out[366]= $\left\{e^{t(\delta-\beta\rho)}\, Z0\right\}$

In[367]:=
```
Co[β_, ρ_, δ_, Z0_, t_] = Factor[Evaluate[ Cs[t] /. sol2 /. C[1] → 0 ]]
Zo[β_, ρ_, δ_, Z0_, t_] = Factor[Evaluate[ Z[t] /. sol2 /. C[1] → 0 ]]
Simplify[Co[0.7, 0.02, 0.5,  1, t]]
```

Out[367]= $\left\{\left\{\dfrac{e^{t(\delta-\beta\rho)}\, Z0\, \beta\rho}{\delta}\right\}\right\}$

Out[368]= $\left\{e^{t(\delta-\beta\rho)}\, Z0\right\}$

Out[369]= $\left\{\left\{0.028\, e^{0.486\, t}\right\}\right\}$

In[370]:=
```
Plot[Co[0.7, 0.02, 0.1,  1, t], {t, 0, 100}, AxesLabel → {"t", "C(t)"}]
```

Out[370]=


In[371]:=
```
Plot[Zo[0.7, 0.02, 0.1, 1, t], {t, 0, 100}, AxesLabel → {"t", "Z(t)"}]
```

Out[371]=